RegularHarmonics

a Mathematica 4.2 package for computing with regular quaternionic functions

Alessandro Perotti
Università degli Studi di Trento

Trento, April 2004

RegularHarmonics

by Alessandro Perotti
Version 1.2 - April 2004

This package implements computations with Fueter-regular quaternionic functions and harmonic functions of two complex variables.

■ Reference

■ *Title*

RegularHarmonics - Version 1.2 - April 2004

■ *Author*

Alessandro Perotti

■ *Summary*

This package implements computations with Fueter-regular quaternionic functions and harmonic functions of two complex variables.

■ *Copyright*

■ *Mathematica* ***Version:*** 4.2

■ *References*

- A. Sudbery, *Quaternionic analysis*, Mat. Proc. Camb. Phil. Soc. vol. 85, 199-225 (1979)
- A. Perotti, *Quaternionic regular functions and the dibar-Neumann problem in C^2,* preprint UTM n.654 (2003) (http://www.science.unitn.it/~perotti/reg.pdf)
- A. Perotti, *A differential criterium for regularity of quaternionic functions,* Comptes Rendus Mathematique, Volume 337, Issue 2, 89-92 (2003)
- http://www.science.unitn.it/~perotti/RegularHarmonics.htm

■ Interface

■ *Initial messages and package context*

```
Off[General::"spell"];Off[General::"spell1"]
Print["RegularHarmonics by A.Perotti, Version 1.2, April 2004"]
Print["This package implements computations with Fueter-regular quaternionic
polynomials and harmonic functions of two complex variables."]
Print["Addional information are available on the world wide web at the page
http://www.science.unitn.it/~perotti/RegularHarmonics.htm"]
Print["Send comments and bug reports to: perotti@science.unitn.it"]

BeginPackage["RegularHarmonics`"]
```

■ *Usage messages*

■ *Error messages*

```
RegularHarmonics::notpoly = "`1` is not a polynomial in `2`"
```

■ Implementation

■ *Begin the private context*

```
Begin["`Private`"]
```

■ *Unprotect system functions*

```
protected = Unprotect[Conjugate,D]
```

■ *Definition of auxiliary functions*

■ Norms

■ ComplexNorm

```
ComplexNorm[a_?NumericQ*z_]:=Abs[a] ComplexNorm[z]
ComplexNorm[Conjugate[z_]]:=ComplexNorm[z]
ComplexNorm[a_?NumericQ]:=Abs[a]
Format[ComplexNorm[z_],StandardForm]:=BracketingBar[z]
```

■ symbolC

```
symbolC[f_, z_Symbol: z] := Module[{}, Format[z[i_], StandardForm] = zi;
  Symbol[SymbolName[z] <> "-"][i_]  := Conjugate[z[i]];
  ReplaceAll[f, {zi_ → z[i] , Subscript[z, i_] → z[i]}] /.
    ComplexNorm[z] → Sqrt[z[1] Conjugate[z[1]] + z[2] Conjugate[z[2]]] /.
   ComplexNorm[z[i_]] → Sqrt[z[i] Conjugate[z[i]]]]]
```

■ ToComplexNorm

```
ToComplexNorm[f_, z_Symbol: z] := Module[{cn},
   Simplify[symbolC[f, z], z[1] Conjugate[z[1]] + z[2] Conjugate[z[2]] == cn] /.
    cn → ComplexNorm[z]^2] /. z[i_]^(n_Integer: 1) Conjugate[z[i_]]^(m_Integer: 1) →
   ComplexNorm[z[i]]^(2 Min[n, m]) z[i]^(n - Min[n, m])
    Conjugate[z[i]]^(m - Min[n, m])
SetAttributes[ToComplexNorm, Listable]
```

■ ToRealNorm

```
ToRealNorm[f_, x_Symbol: x] := Module[{rn},
  Simplify[f /. xi_ → x[i], Sum[x[i]^2, {i, 0, 3}] == rn] /. rn → ComplexNorm[x]^2]
SetAttributes[ToRealNorm, Listable]
```

■ Tonorm

```
Tonorm = False;
ToCxNorm[f_, z_Symbol: z] := If[Tonorm, ToComplexNorm[f, z], f, f]
```

■ Auxiliary functions for polynomials

■ NormalSeries

```
NormalSeries[f_, n_Integer, z_Symbol: z] := Module[{zb, t},
  Normal[Series[symbolC[f, z] /. Conjugate[z[i_]] → t zb[i] /. z[i_] → t z[i] ,
      {t, 0, n}]] /. t → 1 /. zb[i_] -> Conjugate[z[i]]]
```

HomogeneousParts

```
HomogeneousParts[f_, z_Symbol: z] := Module[{e, t, l, k, st},
  e = symbolC[f, z] ;
  If[PolynomialQ[e, {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}],
   t = First[Internal`DistributedTermsList[
      e, {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}]];
   While[Length[t] == 0, t = {{{0, 0, 0, 0}, 0}}]; l = {};
   While[Length[t] > 0,
    k = Plus @@ First[t][[1]];
    st = Select[t, Plus @@ First[#] == k &]; t = Complement[t, st];
    l = Append[l, {Internal`FromDistributedTermsList[
        {st, {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}}], k}]];
   Sort[l, OrderedQ[{#1[[2]], #2[[2]]}] &], Message[
    RegularHarmonics::notpoly, f, z]]]
HomogeneousParts[f_, n_Integer, z_Symbol: z] :=
 HomogeneousParts[NormalSeries[f, n, z], z]
SetAttributes[HomogeneousParts, Listable]
```

ComplexHomogeneousParts

```
ComplexHomogeneousParts[f_, z_Symbol: z] := Module[{e, t, l, k, st},
  e = symbolC[f, z] ;
  If[PolynomialQ[e, {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}],
   t = First[ Internal`DistributedTermsList[e,
      {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}]]; l = {};
   While[Length[t] > 0,
    p = Plus @@ First[t][[1, {1, 2}]]; q = Plus @@ First[t][[1, {3, 4}]];
    st = Select[Select[t, Plus @@ #[[1, {1, 2}]] == p &],
      Plus @@ #[[1, {3, 4}]] == q &]; t = Complement[t, st];
    l = Append[l, {Internal`FromDistributedTermsList[
        {st, {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}}], {p, q}}]];
   Sort[Sort[l], OrderedQ[{Plus @@ #1[[2]], Plus @@ #2[[2]]}] &],
   Message[RegularHarmonics::notpoly, f, z]]]
ComplexHomogeneousParts[f_, n_Integer, z_Symbol: z] :=
 ComplexHomogeneousParts[NormalSeries[f, n, z], z]
SetAttributes[ComplexHomogeneousParts, Listable]
```

TotalDegree

```
TotalDegree[f_, z_Symbol: z] :=
 Exponent[symbolC[f, z] /. Conjugate[z[i_]] → z[1] /. z[2] → z[1], z[1]]
```

LeadingTerm

```
LeadingTerm[f_, z_Symbol: z] :=
 Module[{l}, l = First[Internal`DistributedTermsList[symbolC[f, z] ,
     {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}]];
  l = Sort[Sort[l], OrderedQ[{Plus @@ #1[[1]], Plus @@ #2[[1]]}] &];
  Internal`FromDistributedTermsList[
   {{Last[l]}, {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}}]]
SetAttributes[LeadingTerm, Listable]
```

Miscellaneous

OnS (trace on the boundary of the unit ball)

```
OnS[f_, z_Symbol: z] := Simplify[symbolC[f /. ComplexNorm[z] -> 1, z] ,
  z[1] Conjugate[z[1]] + z[2] Conjugate[z[2]] == 1 ]
```

Definition of principal functions

Laplacian and Kelvin Transform

Laplacian (real)

```
Laplacian[f_, x_Symbol: x] :=
 Simplify[Sum[D[f /. x_j_ -> x[j], {x[j], 2}], {j, 0, 3}] /. x[j_] -> x_j]
```

ComplexLaplacian

```
ComplexLaplacian[f_, z_Symbol: z] :=
 Module[{e}, e = symbolC[f, z]; Simplify[Sum[D[e , Conjugate[z[j]], z[j]], {j, 2}]]]
```

KelvinTransform

```
KelvinTransform[f_, z_Symbol: z] := Module[{hp}, hp = HomogeneousParts[symbolC[f, z]];
  Sum[hp[[i, 1]] ComplexNorm[z]^(-2 - 2 hp[[i, 2]]), {i, Length[hp]}]]
```

Field Conversions

RtoC

```
RtoC[{f1_, f2_}, x_Symbol: x, z_Symbol: z] :=
  (symbolC[0, z]; Expand[symbolC[f1 + I * f2, x] /.
      x[i_?OddQ] → (z[(i + 1) / 2] - Conjugate[z[(i + 1) / 2]]) / (2 I) /.
     x[i_?EvenQ] → (z[(i + 2) / 2] + Conjugate[z[(i + 2) / 2]]) / 2]);
```

CtoR

```
CtoR[f_, z_Symbol: z, x_Symbol: x] :=
 Module[{e}, e = Expand[symbolC[f, z] /. Conjugate[z[i_]] → x[2 i - 2] - ⅈ * x[2 i - 1] /.
     z[i_] → x[2 i - 2] + ⅈ x[2 i - 1]]; ComplexExpand[{Re[e], Im[e]}] /. x[j_] -> x_j]
```

- **CtoH**

```
CtoH[f_List, z_Symbol: z, x_Symbol: x] :=
  Flatten[Table[CtoR[f, z, x][[All, i]], {i, 2}]];
```

- **HtoC**

```
HtoC[{f0_, f1_, f2_, f3_}, x_Symbol: x, z_Symbol: z] :=
  {RtoC[{f0, f1}, x, z], RtoC[{f2, f3}, x, z]};
```

- **Cauchy-Riemann-Fueter equations for regular and ψ-regular functions and related boundary operators**

- **CRF**

```
CRF[{f1_, f2_}, z_Symbol: z] :=
 Module[{e1, e2}, {e1, e2} = Expand[symbolC[{f1, Conjugate[f2]}, z]];
  Simplify[{D[f1, Conjugate[z[1]]] - D[Conjugate[f2], Conjugate[z[2]]],
    D[f1, z[2]] + D[Conjugate[f2], z[1]]}] ]
```

- **PsiCRF**

```
PsiCRF[{f1_, f2_}, z_Symbol: z] :=
 Module[{e1, e2}, {e1, e2} = Expand[symbolC[{f1, Conjugate[f2]}, z]]; Simplify[
   {D[e1, Conjugate[z[1]]] - D[e2, z[2]], D[e1, Conjugate[z[2]]] + D[e2, z[1]]}] ]
```

- **DbarN**

```
DbarN[f_, z_Symbol: z] :=
 Module[{e, zb}, e = Expand[symbolC[f, z] /. Conjugate[z[i_]] → zb[i]];
  Sum[zb[i] D[e, zb[i]], {i, 2}] /. zb[i_] → Conjugate[z[i]] ]
```

- **L**

```
L[f_, z_Symbol: z] := Module[{zp, e}, zp[i_?OddQ] = z[i + 1] ;
  zp[i_?EvenQ] = -z[i - 1]; symbolC[0, z]; Sum[zp[i] D[f, Conjugate[z[i]]], {i, 2}] ]
```

- **Lbar**

```
Lbar[f_, z_Symbol: z] := Module[{zp, zb, e},
  zp[i_?OddQ] = Conjugate[z[i + 1]] ; zp[i_?EvenQ] = -Conjugate[z[i - 1]];
  e = Expand[symbolC[f, z] /. Conjugate[z[i_]] → zb[i]];
  Sum[zp[i] D[e, z[i]], {i, 2}] /. zb[i_] → Conjugate[z[i]]]
```

■ **NFueter**

```
NFueter[f_, z_Symbol: z] :=
 Module[{e, zb}, e = Expand[symbolC[f, z]  /. Conjugate[z[i_]] → zb[i]];
  zb[1] D[e, zb[1]] + z[2] D[e, z[2]] /. zb[i_] → Conjugate[z[i]] ]
```

■ **TFueter**

```
TFueter[f_, z_Symbol: z] :=
 Module[{zb, e}, e = Expand[symbolC[f, z] /. Conjugate[z[i_]] → zb[i]];
  zb[2] D[e, zb[1]] - z[1] D[e, z[2]] /. zb[i_] → Conjugate[z[i]]]
```

■ **PsiRegularQ**

```
PsiRegularQ[{f1_, f2_}, z_Symbol: z] :=
 {OnS[DbarN[f1, z] + Lbar[Conjugate[f2], z], z], ComplexLaplacian[{f1, f2}, z]} ===
  {0, {0, 0}}
```

■ **RegularQ**

```
RegularQ[{f1_, f2_}, z_Symbol: z] :=
 {OnS[NFueter[f1, z] + Conjugate[TFueter[f2, z]], z],
   ComplexLaplacian[{f1, f2}, z]} === {0, {0, 0}}
```

■ **Gauss formulas for harmonic extension and harmonic decomposition of polynomials on the unit ball and on the exterior of the unit ball**

■ **GaussExtension**

```
HomGaussExtension[e_, k_Integer, z_Symbol: z] := Module[{f, k2, lap},
  f = symbolC[e, z] ; k2 = Floor[k / 2];
  lap[0] = f; lap[1] = ComplexLaplacian[f, z];
  Do[lap[j] = ComplexLaplacian[lap[j - 1], z], {j, 2, k2}];
  Sum[(k - 2 s + 1) / (s ! (k - s + 1) !) Sum[(-1) ^j (k - j - 2 s) ! / j !
      Sum[z[i] Conjugate[z[i]], {i, 2}] ^j lap[j + s], {j, 0, k2 - s}], {s, 0, k2}]]
GaussExtension[e_, z_Symbol: z] := Module[{hp}, f = symbolC[e, z];
  If[PolynomialQ[f, {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}],
   hp = HomogeneousParts[f, z];
   ToCxNorm[Expand[Sum[HomGaussExtension[hp[[i, 1]], hp[[i, 2]], z],
      {i, Length[hp]}]], z], Message[RegularHarmonics::notpoly, f, z]]]
GaussExtension[e_, n_Integer, z_Symbol: z] := GaussExtension[NormalSeries[e, n, z], z]
SetAttributes[GaussExtension, Listable]
```

GaussForm

```
HomGaussForm[e_, k_Integer, z_Symbol: z] := Module[{f, k2, lap},
  f = symbolC[e, z] ; k2 = Floor[k / 2];
  lap[0] = f; lap[1] = ComplexLaplacian[f, z];
  Do[lap[j] = ComplexLaplacian[lap[j - 1], z], {j, 2, k2}];
  Table[
   {Sum[(k - 2 s + 1) / (s ! (k - s + 1) !) (-1) ^j (k - j - 2 s) ! / j ! Sum[z[i] Conjugate[z[i]],
        {i, 2}] ^j * lap[j + s], {j, 0, k2 - s}], 2 s}, {s, 0, k2}] ]
GaussForm[e_, z_Symbol: z] := Module[{hp, ghp, st, l, k}, f = symbolC[e, z];
  If[PolynomialQ[f, {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}],
   hp = HomogeneousParts[f, z];
   ghp = Flatten[Table[HomGaussForm[hp[[i, 1]], hp[[i, 2]], z], {i, Length[hp]}], 1];
   l = {};
   While[Length[ghp] > 0, k = First[ghp][[2]];
    st = Select[ghp, #[[2]] == k &]; ghp = Complement[ghp, st];
    l = Append[l, {Plus @@ st[[All, 1]], k}]];
   ToCxNorm[Expand[Sort[l, OrderedQ[{#1[[2]], #2[[2]]}] &]], z],
   Message[RegularHarmonics::notpoly, f, z]]]
GaussForm[e_, n_Integer, z_Symbol: z] := GaussForm[NormalSeries[e, n, z], z]
SetAttributes[GaussForm, Listable]
```

ExteriorGaussExtension

```
ExteriorHomGaussExtension[e_, k_Integer, z_Symbol: z] := Module[{f, k2, lap},
  f = symbolC[e, z] ; k2 = Floor[k / 2];
  lap[0] = f; lap[1] = ComplexLaplacian[f, z];
  Do[lap[j] = ComplexLaplacian[lap[j - 1], z], {j, 2, k2}];
  Sum[(k - 2 s + 1) / (s ! (k - s + 1) !)
    Sum[(-1) ^j (k - j - 2 s) ! / j ! Sum[z[i] Conjugate[z[i]], {i, 2}] ^ (j - k + 2 s - 1)
      lap[j + s], {j, 0, k2 - s}], {s, 0, k2}]]
ExteriorGaussExtension[e_, z_Symbol: z] := Module[{hp}, f = symbolC[e, z];
  If[PolynomialQ[f, {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}],
   hp = HomogeneousParts[f, z];
   ToCxNorm[Expand[Sum[ExteriorHomGaussExtension[hp[[i, 1]], hp[[i, 2]], z],
      {i, Length[hp]}]], z], Message[RegularHarmonics::notpoly, f, z]]]
ExteriorGaussExtension[e_, n_Integer, z_Symbol: z] :=
 ExteriorGaussExtension[NormalSeries[e, n, z], z]
SetAttributes[ExteriorGaussExtension, Listable]
```

Regular and ψ-regular extensions of polynomials on the unit ball

Dk

```
Dk[e_, k_, z_Symbol: z] := Module[{f, k2, lap},
  f = symbolC[e, z] ; k2 = Floor[k / 2];
  lap[0] = f; lap[1] = ComplexLaplacian[f, z];
  Do[lap[j] = ComplexLaplacian[lap[j - 1], z], {j, 2, k2}];
  1 / k ! Sum[2^l (k - 2 l - 1) ! (2 l - 1) !! / (l + 1) ! lap[l + 1], {l, 0, k2 - 1}]]
```

■ PsiRegularExtensionQ

```
PsiRegularExtensionQ[{f1_, f2_}, z_Symbol: z] :=
 Module[{hp}, hp = HomogeneousParts[f1, z]; OnS[DbarN[f1, z] + Lbar[Conjugate[f2], z] -
     Sum[Dk[hp[[i, 1]], hp[[i, 2]], z], {i, Length[hp]}], z] === 0]
```

■ RegularExtensionQ

```
RegularExtensionQ[{f1_, f2_}, z_Symbol: z] :=
 Module[{hp}, hp = HomogeneousParts[f1, z];
  OnS[NFueter[f1, z] + Conjugate[TFueter[f2, z]] -
     Sum[Dk[hp[[i, 1]], hp[[i, 2]], z], {i, Length[hp]}], z] === 0]
```

■ PsiRegularExtension

```
PsiRegularExtension[{f1_, f2_}, z_Symbol: z] :=
 Module[{ge}, ge = GaussExtension[{f1, f2}, z];
  If[PsiRegularQ[ge, z], ge, "No ψ-regular extension"]]
PsiRegularExtension[f1_, z_Symbol: z] :=
 Module[{chp, ge, f2}, chp = ComplexHomogeneousParts[f1, z];
  ge = Table[HomGaussForm[chp[[i, 1]], Plus @@ chp[[i, 2]], z], {i, Length[chp]}];
  f2 = ToCxNorm[Expand[
     Sum[Sum[1 / (First[chp[[i, 2]]] - s + 1) Lbar[Conjugate[First[ge[[i, s + 1]]]], z],
       {s, 0, Min[chp[[i, 2]]]}], {i, Length[chp]}]], z]; {GaussExtension[f1, z], f2}]
PsiRegularExtension[f1_, n_Integer, z_Symbol: z] :=
 PsiRegularExtension[NormalSeries[f1, n, z], z]
```

■ RegularExtension

```
RegularExtension[{f1_, f2_}, z_Symbol: z] := Module[{ge},
  ge = GaussExtension[{f1, f2}, z]; If[RegularQ[ge, z], ge, "No regular extension"]]
RegularExtension[f1_, z_Symbol: z] := Expand[
  PsiRegularExtension[f1 /. z[2] → Conjugate[z[2]], z] /. z[2] → Conjugate[z[2]]]
RegularExtension[f1_, n_Integer, z_Symbol: z] :=
 RegularExtension[NormalSeries[f1, n, z], z]
```

■ Sphere and ball integrals

■ SphereIntegral

```
SphereIntegral[f_List, z_Symbol: z] :=
 {SphereIntegral[f[[1]] , z], SphereIntegral[f[[2]], z]}
SphereIntegral[f_, z_Symbol: z] := Module[{e, t, a},
  e = symbolC[f, z] ;
  If[PolynomialQ[e, {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}],
   t = First[ Internal`DistributedTermsList[e,
       {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}]];
   Sum[If[(a = Part[t, i][[1, {1, 2}]]) == Part[t, i][[1, {3, 4}]],
     Part[t, i][[2]] Times @@ (a !) / (Plus @@ a + 1) !, 0], {i, Length[t]}],
   Message[RegularHarmonics::notpoly, f, z]]]
```

■ SphereProduct

```
SphereProduct[{f1_, f2_}, {g1_, g2_}, z_Symbol: z] := Module[{e1, h1, e2, h2},
  {e1, h1, e2, h2} = symbolC[{f1, g1, f2, g2}, z] ; SphereIntegral[
   {h1 Conjugate[e1] + e2 Conjugate[h2], h2 Conjugate[e1] - e2 Conjugate[h1]}, z]]
SphereProduct[f_, g_, z_Symbol: z] := Module[{f1, g1},
  {f1, g1} = symbolC[{f, g}, z] ; SphereIntegral[f1 Conjugate[g1], z]]
```

■ SphereNorm

```
SphereNorm[{f1_, f2_}, z_Symbol: z] := Module[{e1, e2}, {e1, e2} = symbolC[{f1, f2}, z] ;
  Sqrt[SphereIntegral[e1 Conjugate[e1] + e2 Conjugate[e2], z]]]
SphereNorm[f_, z_Symbol: z] := Module[{e}, e = symbolC[f, z] ;
  Sqrt[SphereIntegral[e Conjugate[e], z]]]
```

■ BallIntegral

```
BallIntegral[f_List, z_Symbol: z] :=
 {BallIntegral[f[[1]] , z], BallIntegral[f[[2]], z]}
BallIntegral[f_, z_Symbol: z] := Module[{e, t, a},
  e = symbolC[f, z] ;
  If[PolynomialQ[e, {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}],
   t = First[ Internal`DistributedTermsList[e,
      {z[1], z[2], Conjugate[z[1]], Conjugate[z[2]]}]];
   Sum[If[(a = Part[t, i][[1, {1, 2}]]) == Part[t, i][[1, {3, 4}]],
     2 Part[t, i][[2]] Times @@ (a !) / (Plus @@ a + 2) !, 0], {i, Length[t]}],
   Message[RegularHarmonics::notpoly, f, z]]]
```

■ BallProduct

```
BallProduct[{f1_, f2_}, {g1_, g2_}, z_Symbol: z] :=
 Module[{e1, h1, e2, h2}, {e1, h1, e2, h2} = symbolC[{f1, g1, f2, g2}, z] ; BallIntegral[
   {h1 Conjugate[e1] + e2 Conjugate[h2], h2 Conjugate[e1] - e2 Conjugate[h1]}, z]]
BallProduct[f_, g_, z_Symbol: z] := Module[{f1, g1},
  {f1, g1} = symbolC[{f, g}, z] ; BallIntegral[f1 Conjugate[g1], z]]
```

■ BallNorm

```
BallNorm[{f1_, f2_}, z_Symbol: z] := Module[{e1, e2}, {e1, e2} = symbolC[{f1, f2}, z] ;
  Sqrt[BallIntegral[e1 Conjugate[e1] + e2 Conjugate[e2], z]]]
BallNorm[f_, z_Symbol: z] := Module[{e}, e = symbolC[f, z] ;
  Sqrt[BallIntegral[e Conjugate[e], z]]]
```

Spherical harmonics bases

BasisP

```
CBasisP[p_, q_, l_, r_] := (-1)^r Binomial[p, l - r] Binomial[q, r]
BasisP[p_Integer, q_Integer, z_Symbol: z] := Module[{t}, symbolC[0, z];
  t = Table[Sum[CBasisP[p, q, l, r] z[1]^(p - l + r) z[2]^(l - r) Conjugate[z[1]]^r
      Conjugate[z[2]]^(q - r), {r, Max[l - p, 0], Min[q, l]}], {l, 0, Floor[(p + q) / 2]}];
  Join[If[EvenQ[p + q], Delete[t, -1], t], Reverse[t] /. {z[1] → z[2], z[2] -> z[1]}]]
BasisP[k_Integer, z_Symbol: z] := Module[{t},
  t = Table[BasisP[k - q, q, z], {q, 0, Floor[k / 2]}]; Flatten[
   Join[If[EvenQ[k], Delete[t, -1], t], Reverse[t] /. z[i_] → Conjugate[z[i]]]]]
```

RealBasisP

```
RealBasisP[k_Integer, z_Symbol: z] :=
 Module[{t}, t = Table[BasisP[k - q, q, z], {q, 0, Floor[k / 2]}]; t = Join[Delete[
     Flatten[t + Conjugate[t]], If[EvenQ[k], Table[{i}, {i, -k / 2, -1}], {}]], Delete[
     Flatten[i (-t + Conjugate[t])], If[EvenQ[k], Table[{i}, {i, -k / 2 - 1, -1}], {}]]];
  Expand[If[EvenQ[k], ReplacePart[t, t[[k^2 / 2 + k + 1]] / 2, k^2 / 2 + k + 1], t]]]
```

ONBasisP

```
ONBasisP[p_Integer, q_Integer, z_Symbol: z] := Module[{c, t}, symbolC[0, z];
  Do[c[l, r] = CBasisP[p, q, l, r],
   {l, 0, Floor[(p + q) / 2]}, {r, Max[0, l - p], Min[q, l]}];
  t = Table[Sum[c[l, r] z[1]^(p - l + r) z[2]^(l - r) Conjugate[z[1]]^r
       Conjugate[z[2]]^(q - r), {r, Max[l - p, 0], Min[q, l]}]
     Sqrt[(p + q + 1) !] / Sqrt[ Sum[c[l, s] Sum[c[l, r] (p - l + r + s) ! (q + l - r - s) !,
         {r, Max[0, l - p], Min[q, l]}], {s, Max[0, l - p], Min[q, l]}]],
    {l, 0, Floor[(p + q) / 2]}]; Join[If[EvenQ[p + q], Delete[t, -1], t],
   Reverse[t] /. {z[1] → z[2], z[2] -> z[1]}]]
ONBasisP[k_Integer, z_Symbol: z] := Module[{t},
  t = Table[ONBasisP[k - q, q, z], {q, 0, Floor[k / 2]}]; Flatten[
   Join[If[EvenQ[k], Delete[t, -1], t], Reverse[t] /. z[i_] → Conjugate[z[i]]]]]
```

BallONBasisP

```
BallONBasisP[p_Integer, q_Integer, z_Symbol: z] := ONBasisP[p, q, z] Sqrt[(p + q + 2) / 2]
BallONBasisP[k_Integer, z_Symbol: z] := ONBasisP[k, z] Sqrt[(k + 2) / 2]
RealONBasisP[k_Integer, z_Symbol: z] :=
 Module[{t}, t = Table[ONBasisP[k - q, q, z] / Sqrt[2], {q, 0, Floor[k / 2]}];
  t = Join[Delete[Flatten[t + Conjugate[t]],
     If[EvenQ[k], Table[{i}, {i, -k / 2, -1}], {}]], Delete[
     Flatten[i (-t + Conjugate[t])], If[EvenQ[k], Table[{i}, {i, -k / 2 - 1, -1}], {}]]];
  Expand[If[EvenQ[k], ReplacePart[t, t[[k^2 / 2 + k + 1]] / Sqrt[2], k^2 / 2 + k + 1], t]]]
RealBallONBasisP[k_Integer, z_Symbol: z] := RealONBasisP[k, z] Sqrt[(k + 2) / 2]
BasisG[p_Integer, q_Integer, α_: α, z_Symbol: z] :=
 (symbolC[0, z]; (z[1] + α z[2])^p (Conjugate[z[2]] - α Conjugate[z[1]])^q)
```

■ Regular and ψ-regular spherical harmonics bases

■ PsiRegularBasis

```
PsiRegularBasis[k_?EvenQ, z_Symbol: z] :=
 Module[{b}, Do[b[p, k - p] = BasisP[p, k - p, z], {p, k / 2, k}];
  Expand[Flatten[Table[{b[p, k - p][[i]],
      (p + 1) ^ (-1) Conjugate[L[b[p, k - p][[i]], z]]}, {p, k, k / 2, -1}, {i, k + 1}], 1]]]
PsiRegularBasis[k_?OddQ, z_Symbol: z] := Module[{b},
  Do[b[p, k - p] = BasisP[p, k - p, z], {p, (k - 1) / 2, k}]; Expand[Delete[
    Flatten[Table[{b[p, k - p][[i]], (p + 1) ^ (-1) Conjugate[L[b[p, k - p][[i]], z]]},
      {p, k, (k - 1) / 2, -1}, {i, k + 1}], 1], Table[{-i}, {i, (k + 1) / 2}]]]]
```

■ PsiRegularONBasis

```
PsiRegularONBasis[k_?EvenQ, z_Symbol: z] :=
 Module[{b}, Do[b[p, k - p] = ONBasisP[p, k - p, z], {p, k / 2, k}]; Expand[
   Flatten[Table[{b[p, k - p][[i]], (p + 1) ^ (-1) Conjugate[L[b[p, k - p][[i]], z]]}
      Sqrt[(p + 1) / (k + 1)], {p, k, k / 2, -1}, {i, k + 1}], 1]]]
PsiRegularONBasis[k_?OddQ, z_Symbol: z] := Module[{b},
  Do[b[p, k - p] = ONBasisP[p, k - p, z], {p, (k - 1) / 2, k}];
  Expand[Delete[Flatten[Table[{b[p, k - p][[i]],
        (p + 1) ^ (-1) Conjugate[L[b[p, k - p][[i]], z]]} Sqrt[(p + 1) / (k + 1)],
      {p, k, (k - 1) / 2, -1}, {i, k + 1}], 1], Table[{-i}, {i, (k + 1) / 2}]]]]
```

■ PsiRegularBallONBasis

```
PsiRegularBallONBasis[k_Integer, z_Symbol: z] :=
 PsiRegularONBasis[k, z] Sqrt[(k + 2) / 2]
```

■ RegularBasis

```
RegularBasis[k_Integer, z_Symbol: z] :=
 Expand[PsiRegularBasis[k, z] /. z[2] → Conjugate[z[2]]]
```

■ RegularONBasis

```
RegularONBasis[k_Integer, z_Symbol: z] :=
 Expand[PsiRegularONBasis[k, z] /. z[2] → Conjugate[z[2]]]
```

■ RegularBallONBasis

```
RegularBallONBasis[k_Integer, z_Symbol: z] :=
 Expand[PsiRegularBallONBasis[k, z] /. z[2] → Conjugate[z[2]]]
```

New definitions for system functions

```
Conjugate[z_Plus]:=Conjugate/@z
Conjugate[z_Times]:=Conjugate/@z
Conjugate[z_^n_Integer]:=Conjugate[z]^n
Conjugate[Conjugate[z_]]:=z
Conjugate'[z_]:=0
Format[Conjugate[z_],StandardForm]:=OverBar[z]

D[f_,Conjugate[z_]]:=Conjugate[D[Conjugate[f],z]]

(*MakeExpression[RowBox[{"OverBar","[",x_,"]"}],FullForm]:=MakeExpression[RowBox[{"
Conjugate","[",x,"]"}],FullForm]*)
```

Restore protection of system symbols

```
Protect[ Evaluate[protected] ]
```

End the private context

```
End[ ]
```

Epilog

Protect exported symbol

```
Protect[Evaluate[$Context <> "*"]]
```

End the package context

```
EndPackage[ ]
```

Using RegularHarmonics

Alessandro Perotti - Version 1.2 - April 2004

RegularHarmonics is a *Mathematica 4.2* package for making computations with Fueter-regular quaternionic functions and harmonic functions of two complex variables. It is based on the results obtained in [*S*],[*P1*] and [*P2*].

Addional information are available on the world wide web at the page *http : // www.science.unitn.it / ~perotti / regular harmonics.htm*

Please send comments and bug reports to: perotti@science.unitn.it.

Loading the package

To use the **RegularHarmonics** package, you have to load it with the command **<<** (or equivalently with **`Get`**) followed by the name of the .m file. You can use the menu command **Input/Get File Path** to search for and paste the full pathname of the file RegularHarmonics.m.

```
<< "C:\\...\\RegularHarmonics.m"
```

```
RegularHarmonics by A.Perotti, Version 1.2, April 2004

This package implements computations with Fueter-regular quaternionic
  polynomials and harmonic functions of two complex variables.

Addional information are available on the world wide web at the
  page http://www.science.unitn.it/~perotti/RegularHarmonics.htm

Send comments and bug reports to: perotti@science.unitn.it
```

Default variables

The symbol **`z`** denotes the default indexed complex variable in C^2, with two components **`z[1]`**, **`z[2]`**.
t is identified with the quaternion $z_1 + z_2\, j$. The
complex conjugate **`Conjugate[z[1]]`** can be input as **`z-[1]`**
(the conjugation character is obtained with the sequence ESC-ESC) and is output as $\overline{z_1}$.The same holds for $\overline{z_2}$.

The symbol **x** denotes the default indexed real variable with four components **x[0], x[1], x[2], x[3]**. It represents the quaternion $x_0 + i\, x_1 + j\, x_2 + k\, x_3$ and the complex pair $(z_1, z_2) = (x_0 + i\, x_1, x_2 + i\, x_3)$.

Laplacian

Laplacian[f,x] gives the (ordinary) Laplacian of **f** with respect to **x**.

```
Laplacian[x[1]^2 x[3]^3]
```

$2\, x_3\, (3\, x_1^2 + x_3^2)$

ComplexLaplacian[f, z] gives the complex Laplacian of **f** with respect to **z**. In $\mathbf{C}^2$ it is equal to 1 / 4 of the real Laplacian of **f**.

```
ComplexLaplacian[z[1]^2 z-[1] + z[2]^3]
```

$2\, z_1$

Field conversions

The following functions perform two-ways conversions between real, complex and quaternionic fields.

RtoC[{g1, g2}, x, z] converts the real pair $\{g_1, g_2\}$ as a function of **x** to the complex expression $g_1 + i\, g_2$ as a function of **z**.

```
cx = RtoC[{x[0] x[1], x[2]}]
```

$$\frac{1}{4}\, i\, \overline{z_1}^2 + \frac{i\, \overline{z_2}}{2} - \frac{i\, z_1^2}{4} + \frac{i\, z_2}{2}$$

Variables different from the defaults can be given explicitely.

```
RtoC[{y[0], y[1]}, y, w]
```

w_1

CtoR[f,z] converts a complex expression **f** as a function of **z** to the form {real part, imaginary part} as a function of **x**.

```
CtoR[cx]
```

$\{x_0\, x_1,\ x_2\}$

CtoH[{**f**$_1$, **f**$_2$}, **z**, **x**] converts the pair {**f**$_1$, **f**$_2$} as a complex function of **z** to the 4 – tuple of the real components of the quaternionic expression **f**$_1$ + **f**$_2$ **j**.

```
quat = CtoH[{z[1], z[2] z–[2]}]
```

$\{x_0,\ x_1,\ x_2^2 + x_3^2,\ 0\}$

HtoC[{**g**$_0$, **g**$_1$, **g**$_2$, **g**$_3$}, **x**, **z**]
con verts the 4 – tuple $\{g_0,\ g_1,\ g_2,\ g_3\}$ of the real components of a quaternion as a function of **x** to the complex pair $\{g_0 + i\, g_1,\ g_2 + i\, g_3\}$ as a function of **z**.

```
HtoC[quat]
```

$\{z_1,\ \overline{z_2}\, z_2\}$

Cauchy-Riemann-Fueter equations for regular and ψ-regular functions and related boundary operators

We refer to [P1] and [P2] for the relevant definitions concerning regular, ψ-regular quaterninic functions, the Cauchy-Riemann-Fueter equations and the boundary differential conditions characterizing regular functions on a domain in $\boldsymbol{C}^2$ among harmonic functions.

CRF[{**f**$_1$, **f**$_2$}, **z**] computes the (left) Cauchy – Riemann – Fueter equations of **f** = **f**$_1$ + **f**$_2$ **j** i.e. the pair $\left\{\frac{\partial f_1}{\partial \overline{z_1}} - \frac{\partial \overline{f_2}}{\partial \overline{z_2}},\ \frac{\partial f_1}{\partial z_2} + \frac{\partial \overline{f_2}}{\partial z_1}\right\}$.

```
CRF[{z[1] z[2] z–[2], z[1] + z[2] z–[2]}]
```

$\{-z_2,\ \overline{z_2}\, z_1\}$

PsiCRF[{f_1, f_2}, z] computes the (left) Cauchy – Riemann –
Fueter equations for (left) ψ – regular functions i.e. the pair $\left\{\frac{\partial f_1}{\partial \overline{z_1}} - \frac{\partial \overline{f_2}}{\partial z_2}, \frac{\partial f_1}{\partial \overline{z_2}} + \frac{\partial \overline{f_2}}{\partial z_1}\right\}$.
Note that holomorphic maps of two complex variables define a ψ – regular function.

```
PsiCRF[{z[1] z[2] z-[2], z[1] + z[2] z-[2]}]
```

$\{-\overline{z_2}, z_1 z_2\}$

```
PsiCRF[{z[1]^2 z[2]^3, Sin[z[2]] + 3 z[1] z[2]^2}]
```

{0, 0}

The following five differential operators will be used to give boundary differential conditions characterizing regular and ψ-regular functions on the unit ball in $\boldsymbol{C}^2$ among harmonic functions. Cf. [P1] for details.

DbarN[f, z] gives the normal part $\overline{\partial_n}\, f = \overline{z_1}\, \frac{\partial f}{\partial \overline{z_1}} + \overline{z_2}\, \frac{\partial f}{\partial \overline{z_2}}$ of $\bar{\partial} f$ with respect to the unit sphere S.

```
DbarN[z[1] z[2] z-[2]]
```

$\overline{z_2}\, z_1 z_2$

L[f, z] applies the Cauchy – Riemann tangential (with respect to the unit sphere S) operator $z_2 \frac{\partial}{\partial \overline{z_1}} - z_1 \frac{\partial}{\partial \overline{z_2}}$ to the complex function f.

```
L[z[1] z[2] z-[2]]
```

$-z_1^2 z_2$

Lbar[f, z] applies the conjugate Cauchy –
Riemann tangential (with respect to the unit sphere S) operator $\overline{z_2} \frac{\partial}{\partial z_1} - \overline{z_1} \frac{\partial}{\partial z_2}$ to the complex function f.

```
Lbar[z[1] z[2] z-[2]]
```

$-\overline{z_1}\, \overline{z_2}\, z_1 + \overline{z_2}^2 z_2$

NFueter[**f**, **z**] applies the differential operator $\mathbf{N} = \overline{z_1} \frac{\partial}{\partial \overline{z_1}} + z_2 \frac{\partial}{\partial z_2}$ to **f**.

```
NFueter[z[1] z[2] z-[2]]
```

$\overline{z_2}\, z_1\, z_2$

TFueter[**f**, **z**] applies the tangential operator $\mathbf{T} = \overline{z_2} \frac{\partial}{\partial \overline{z_1}} - z_1 \frac{\partial}{\partial z_2}$ to **f**.

```
TFueter[z[1] z[2] z-[2]]
```

$-\overline{z_2}\, z_1^2$

RegularQ[{f_1, f_2}, **z**] tests for (left) Fueter – regularity of **f** = $f_1 + f_2\, j$ on the unit ball B. Here **f** is a function of **z**, $\overline{z}$.

```
RegularQ[{Sin[z[1]], z[1]^2 z[2]}]
```

```
False
```

PsiRegularQ[{f_1, f_2}, **z**] tests for (left) ψ – regularity of **f** = $f_1 + f_2\, j$ on the unit ball B. Here **f** is a function of **z**, $\overline{z}$.

```
PsiRegularQ[{Sin[z[1]], z[1]^2 z[2]}]
```

```
True
```

Gauss formulas for harmonic extension and harmonic representation of polynomials

GaussForm[**f**, **z**] gives the harmonic representation of the restriction of the polynomial **f**[**z**, $\overline{z}$] to the unit sphere S. The output is a list of pairs {h_k, 2 k}, with h_k harmonic and such that the sum $\sum_k h_k \; | z |^{2k}$ is equal to **f** on S. The applied formula can be found for example in the book *Introduction to the theory of cubature formulas* by S.L. Sobolev.

```
GaussForm[f]
```

$$\left\{\left\{\frac{1}{3}\,\overline{z_1}\,z_1^2-\frac{2}{3}\,\overline{z_2}\,z_1\,z_2+\frac{3}{2}\,\overline{z_1}\,z_1\,z_2^2-\frac{1}{2}\,\overline{z_2}\,z_2^3,\ 0\right\},\ \left\{\frac{2\,z_1}{3}-\frac{3\,z_2^2}{2},\ 2\right\},\ \{0,\ 4\}\right\}$$

GaussExtension[f, z] gives the (polynomial) harmonic extension of the restriction of the polynomial **f[z, z̄]** to the unit sphere S. It is based on the harmonic representation of **f** (see above the function **GaussForm[f, z]**).

```
f := z[1]^2 z-[1] - 2 z[2]^3 z-[2]; ff := GaussExtension[f]; ff
```

$$\frac{2\,z_1}{3}+\frac{1}{3}\,\overline{z_1}\,z_1^2-\frac{2}{3}\,\overline{z_2}\,z_1\,z_2-\frac{3\,z_2^2}{2}+\frac{3}{2}\,\overline{z_1}\,z_1\,z_2^2-\frac{1}{2}\,\overline{z_2}\,z_2^3$$

The restriction of the polynomial to the unit sphere S can be computed by means of the function **OnS**.

```
OnS[ff - f]
```

```
0
```

ExteriorGaussExtension[f, z] gives the harmonic extension on the complement of the unit ball of the restriction of the polynomial **f[z, z̄]** to the unit sphere S.

```
ToComplexNorm[ExteriorGaussExtension[f]]
```

$$\frac{|z|^2\,(2\,|z|^2\,z_1+4\,|z|^4\,z_1+9\,z_2^2-9\,|z|^2\,z_2^2)-6\,\overline{z_2}\,(|z|^2\,z_1\,z_2+2\,z_2^3)}{6\,|z|^{10}}$$

Regular and ψ-regular extensions of polynomials on the unit ball

The following functions use the results given in [P1] and [P2] in order to obtain regular and ψ-regular extension of polynomials.

RegularExtension[{h_1, h_2}, z] and **PsiRegularExtension[{h_1, h_2}, z]** give, if they exist, the (left) regular and the ψ – regular extension **f** = $f_1 + f_2\,j$ of the restriction of **h** = $h_1 + h_2\,j$ to the unit sphere.

RegularExtension[h_1, z] and **PsiRegularExtension[h_1, z]** gives a regular or ψ – regular polynomial **f** = $f_1 + f_2\,j$ such that $f_1 = h_1$ on the unit sphere. Here f_1, f_2 and h_1 must be polynomial functions of **z**, **z̄**. The output is the pair of complex components {f_1, f_2} of **f**.

```
RegularExtension[{z[1]^4 z-[1]^3, z[1]}]
```

```
No regular extension
```

```
ff = RegularExtension[z[1]^4 z-[1]^3]
```

$$\Big\{\frac{2 z_1}{5} + \frac{2}{5}\bar{z}_1 z_1^2 + \frac{6}{35}\bar{z}_1^2 z_1^3 + \frac{1}{35}\bar{z}_1^3 z_1^4 - \frac{4}{5}\bar{z}_2 z_1 z_2 - \frac{36}{35}\bar{z}_1\bar{z}_2 z_1^2 z_2 - \frac{12}{35}\bar{z}_1^2\bar{z}_2 z_1^3 z_2 + \frac{18}{35}\bar{z}_2^2 z_1 z_2^2 + \frac{18}{35}\bar{z}_1\bar{z}_2^2 z_1^2 z_2^2 - \frac{4}{35}\bar{z}_2^3 z_1 z_2^3,\ \frac{2}{5}\bar{z}_1^2 z_2 + \frac{12}{35}\bar{z}_1^3 z_1 z_2 + \frac{3}{35}\bar{z}_1^4 z_1^2 z_2 - \frac{18}{35}\bar{z}_1^2\bar{z}_2 z_2^2 - \frac{12}{35}\bar{z}_1^3\bar{z}_2 z_1 z_2^2 + \frac{6}{35}\bar{z}_1^2\bar{z}_2^2 z_2^3\Big\}$$

```
OnS[ff]
```

$$\Big\{-z_1(-1+\bar{z}_2 z_2)^3,\ \frac{1}{35}\bar{z}_1^2 z_2\,(29 - 48\bar{z}_2 z_2 + 21\bar{z}_2^2 z_2^2)\Big\}$$

```
PsiRegularExtension[{z[1]^4 z-[1]^3, z[1]}]
```

```
No ψ-regular extension
```

```
g = PsiRegularExtension[z-[1] z[1]^2]
```

$$\Big\{\frac{2 z_1}{3} + \frac{1}{3}\bar{z}_1 z_1^2 - \frac{2}{3}\bar{z}_2 z_1 z_2,\ \frac{1}{3}\bar{z}_1^2\bar{z}_2\Big\}$$

The function **CtoH** can be applied to get the four real component of the quaternionic function whose complex components have been computed above.

```
CtoH[g]
```

$$\Big\{\frac{2 x_0}{3} + \frac{x_0^3}{3} + \frac{1}{3}x_0 x_1^2 - \frac{2}{3}x_0 x_2^2 - \frac{2}{3}x_0 x_3^2,\ \frac{2 x_1}{3} + \frac{1}{3}x_0^2 x_1 + \frac{x_1^3}{3} - \frac{2}{3}x_1 x_2^2 - \frac{2}{3}x_1 x_3^2,\ \frac{1}{3}x_0^2 x_2 - \frac{1}{3}x_1^2 x_2 - \frac{2}{3}x_0 x_1 x_3,\ -\frac{2}{3}x_0 x_1 x_2 - \frac{1}{3}x_0^2 x_3 + \frac{1}{3}x_1^2 x_3\Big\}$$

Sphere and ball products and norms

SphereIntegral[**f**, **z**] and **BallIntegral**[**f**, **z**] give the normalized integral over the unit sphere S (resp. the unit ball B) of the polynomial **f**[**z**, **z̄**]. The volume of S and B are normalized to 1.

```
SphereIntegral[z[1]^3 z-[1]^3 z[2] z-[2]]
```

$$\frac{1}{20}$$

```
BallIntegral[z[1]^3 z-[1]^3 z[2] z-[2]]
```

$$\frac{1}{60}$$

SphereProduct[**f**, **g**, **z**] and **BallProduct**[**f**, **g**, **z**] give the normalized L^2 product over the unit sphere S (resp. the unit ball B) of the complex polynomials **f**[**z**, **z̄**] and **g**[**z**, **z̄**].

SphereProduct[{f_1, f_2}, {g_1, g_2}, **z**] and **BallProduct**[{f_1, f_2}, {g_1, g_2}, **z**] give the normalized L^2 product over the unit sphere S (resp. the unit ball B) of the quaternionic polynomials $f_1 + f_2\, j$ and $g_1 + g_2\, j$.

SphereNorm[**f**, **z**] and **BallNorm**[**f**, **z**] give the normalized L^2 norm of the polynomial **f**[**z**, **z̄**].

SphereNorm[{f_1, f_2}, **z**] and **BallNorm**[{f_1, f_2}, **z**] gives the normalized L^2 norm of the quaternionic polynomial $f_1 + f_2\, j$.

```
SphereNorm[RegularExtension[z[1]^3 z-[1]]]
```

$$\frac{\sqrt{13}}{8}$$

```
BallNorm[RegularExtension[z[1]^3 z-[1]]]
```

$$\frac{\sqrt{\frac{19}{3}}}{8}$$

Spherical harmonics bases

BasisP[**p, q, z**] gives a basis of the space $\mathcal{H}_{\mathbf{p,q}}$ of the complex harmonic homogeneous polynomials of degree **p** in $\mathbf{z_1, z_2}$ and **q** in $\overline{\mathbf{z_1}}, \overline{\mathbf{z_2}}$.
It is a $L^2(S)$ – orthogonal basis introduced by Sudbery (see References).

BasisP[**k, z**] gives a basis of the space $\mathcal{H}_{\mathbf{k}} = \bigoplus \mathcal{H}_{\mathbf{p,q}}$ of the complex harmonic homogeneous polynomials of degree **k**.

```
BasisP[2, 3]
```

$$\{\overline{z_2}^3 z_1^2,\ -3\,\overline{z_1}\,\overline{z_2}^2 z_1^2 + 2\,\overline{z_2}^3 z_1 z_2,\ 3\,\overline{z_1}^2\,\overline{z_2}\, z_1^2 - 6\,\overline{z_1}\,\overline{z_2}^2 z_1 z_2 + \overline{z_2}^3 z_2^2,\ \overline{z_1}^3 z_1^2 - 6\,\overline{z_1}^2\,\overline{z_2}\, z_1 z_2 + 3\,\overline{z_1}\,\overline{z_2}^2 z_2^2,\ 2\,\overline{z_1}^3 z_1 z_2 - 3\,\overline{z_1}^2\,\overline{z_2}\, z_2^2,\ \overline{z_1}^3 z_2^2\}$$

RealBasisP[**k, z**] gives a real basis of the space $\mathcal{H}_{\mathbf{k}}$ of the complex harmonic homogeneous polynomials of degree **k** in **z**.

```
b = RealBasisP[3]
```

$$\{\overline{z_1}^3 + z_1^3,\ 3\,\overline{z_1}^2\,\overline{z_2} + 3 z_1^2 z_2,\ 3\,\overline{z_1}\,\overline{z_2}^2 + 3 z_1 z_2^2,\ \overline{z_2}^3 + z_2^3,\ \overline{z_2}\, z_1^2 + \overline{z_1}^2 z_2,\ -\overline{z_1}^2 z_1 - \overline{z_1}\, z_1^2 + 2\,\overline{z_1}\,\overline{z_2}\, z_2 + 2\,\overline{z_2}\, z_1 z_2,\ 2\,\overline{z_1}\,\overline{z_2}\, z_1 - \overline{z_2}^2 z_2 + 2\,\overline{z_1}\, z_1 z_2 - \overline{z_2}\, z_2^2,\ \overline{z_2}^2 z_1 + \overline{z_1}\, z_2^2,\ i\,\overline{z_1}^3 - i\, z_1^3,\ 3\,i\,\overline{z_1}^2\,\overline{z_2} - 3\,i\, z_1^2 z_2,\ 3\,i\,\overline{z_1}\,\overline{z_2}^2 - 3\,i\, z_1 z_2^2,\ i\,\overline{z_2}^3 - i\, z_2^3,\ -i\,\overline{z_2}\, z_1^2 + i\,\overline{z_1}^2 z_2,\ -i\,\overline{z_1}^2 z_1 + i\,\overline{z_1}\, z_1^2 + 2\,i\,\overline{z_1}\,\overline{z_2}\, z_2 - 2\,i\,\overline{z_2}\, z_1 z_2,\ 2\,i\,\overline{z_1}\,\overline{z_2}\, z_1 - i\,\overline{z_2}^2 z_2 - 2\,i\,\overline{z_1}\, z_1 z_2 + i\,\overline{z_2}\, z_2^2,\ i\,\overline{z_2}^2 z_1 - i\,\overline{z_1}\, z_2^2\}$$

```
CtoR[b][[1]]
```

$$\{2 x_0^3 - 6 x_0 x_1^2,\ 6 x_0^2 x_2 - 6 x_1^2 x_2 - 12 x_0 x_1 x_3,\ 6 x_0 x_2^2 - 12 x_1 x_2 x_3 - 6 x_0 x_3^2,\ 2 x_2^3 - 6 x_2 x_3^2,\ 2 x_0^2 x_2 - 2 x_1^2 x_2 + 4 x_0 x_1 x_3,\ -2 x_0^3 - 2 x_0 x_1^2 + 4 x_0 x_2^2 + 4 x_0 x_3^2,\ 4 x_0^2 x_2 + 4 x_1^2 x_2 - 2 x_2^3 - 2 x_2 x_3^2,\ 2 x_0 x_2^2 + 4 x_1 x_2 x_3 - 2 x_0 x_3^2,\ 6 x_0^2 x_1 - 2 x_1^3,\ 12 x_0 x_1 x_2 + 6 x_0^2 x_3 - 6 x_1^2 x_3,\ 6 x_1 x_2^2 + 12 x_0 x_2 x_3 - 6 x_1 x_3^2,\ 6 x_2^2 x_3 - 2 x_3^3,\ 4 x_0 x_1 x_2 - 2 x_0^2 x_3 + 2 x_1^2 x_3,\ -2 x_0^2 x_1 - 2 x_1^3 + 4 x_1 x_2^2 + 4 x_1 x_3^2,\ 4 x_0^2 x_3 + 4 x_1^2 x_3 - 2 x_2^2 x_3 - 2 x_3^3,\ -2 x_1 x_2^2 + 4 x_0 x_2 x_3 + 2 x_1 x_3^2\}$$

ONBasisP[**p, q, z**] gives a $L^2(S)$ – orthonormal basis of the space $\mathcal{H}_{\mathbf{p,q}}$. **ONBasisP**[**k, z**] gives a $L^2(S)$ – orthonormal basis of the space $\mathcal{H}_{\mathbf{k}}$.

BallONBasisP[p, q, z] gives a L^2 (B) – orthonormal basis of the space $\mathcal{H}_{\mathbf{p,q}}$. **BallONBasisP[k, z]** gives a L^2 (B) – orthonormal basis of the space $\mathcal{H}_{\mathbf{k}}$.

```
ONBasisP[2, 3]
```

$$\{2\sqrt{15}\,\overline{z_2}^3 z_1^2,\ 2\sqrt{3}\,(-3\,\overline{z_1}\,\overline{z_2}^2 z_1^2 + 2\,\overline{z_2}^3 z_1 z_2),\ \sqrt{6}\,(3\,\overline{z_1}^2\,\overline{z_2}\, z_1^2 - 6\,\overline{z_1}\,\overline{z_2}^2 z_1 z_2 + \overline{z_2}^3 z_2^2),\ \sqrt{6}\,(\overline{z_1}^3 z_1^2 - 6\,\overline{z_1}^2\,\overline{z_2}\, z_1 z_2 + 3\,\overline{z_1}\,\overline{z_2}^2 z_2^2),\ 2\sqrt{3}\,(2\,\overline{z_1}^3 z_1 z_2 - 3\,\overline{z_1}^2\,\overline{z_2}\, z_2^2),\ 2\sqrt{15}\,\overline{z_1}^3 z_2^2\}$$

```
BallONBasisP[2, 3]
```

$$\{\sqrt{210}\,\overline{z_2}^3 z_1^2,\ \sqrt{42}\,(-3\,\overline{z_1}\,\overline{z_2}^2 z_1^2 + 2\,\overline{z_2}^3 z_1 z_2),\ \sqrt{21}\,(3\,\overline{z_1}^2\,\overline{z_2}\, z_1^2 - 6\,\overline{z_1}\,\overline{z_2}^2 z_1 z_2 + \overline{z_2}^3 z_2^2),\ \sqrt{21}\,(\overline{z_1}^3 z_1^2 - 6\,\overline{z_1}^2\,\overline{z_2}\, z_1 z_2 + 3\,\overline{z_1}\,\overline{z_2}^2 z_2^2),\ \sqrt{42}\,(2\,\overline{z_1}^3 z_1 z_2 - 3\,\overline{z_1}^2\,\overline{z_2}\, z_2^2),\ \sqrt{210}\,\overline{z_1}^3 z_2^2\}$$

Regular and ψ-regular spherical harmonics bases

RegularBasis[k, z] gives a basis of the right quaternionic module $\mathcal{U}_{\mathbf{k}}$ of the (left) regular homogeneous polynomials of degree **k** in **z**.

PsiRegularBasis[k, z] gives a basis of the right quaternionic module $\mathcal{U}_{\mathbf{k}}^{\psi}$ of the (left) ψ – regular homogeneous polynomials of degre e **k** in **z**.

The restrictions to S gives a basis of the *regular harmonics*.

RegularONBasis[k, z] and **RegularBallONBasis[k, z]** give orthonormal bases of the right quaternionic module $\mathcal{U}_{\mathbf{k}}$.

PsiRegularONBasis[k, z] and **PsiRegularBallONBasis[k, z]** give orthonormal bases of the right quaternionic module $\mathcal{U}_{\mathbf{k}}^{\psi}$.

```
RegularBasis[4]
```

$$\left\{\{z_1^4, 0\}, \{4\,\overline{z_2}\,z_1^3, 0\}, \{6\,\overline{z_2}^2 z_1^2, 0\}, \{4\,\overline{z_2}^3 z_1, 0\}, \{\overline{z_2}^4, 0\}, \left\{z_1^3 z_2, -\frac{\overline{z_1}^4}{4}\right\}, \{-\overline{z_1}\,z_1^3 + 3\,\overline{z_2}\,z_1^2 z_2, -\overline{z_1}^3 z_2\}, \left\{3\,\overline{z_1}\,\overline{z_2}\,z_1^2 - 3\,\overline{z_2}^2 z_1 z_2, \frac{3}{2}\,\overline{z_1}^2 z_2^2\right\}, \{3\,\overline{z_1}\,\overline{z_2}^2 z_1 - \overline{z_2}^3 z_2, \overline{z_1}\,z_2^3\}, \left\{\overline{z_1}\,\overline{z_2}^3, \frac{z_2^4}{4}\right\}, \left\{z_1^2 z_2^2, -\frac{2}{3}\,\overline{z_1}^3\,\overline{z_2}\right\}, \left\{-2\,\overline{z_1}\,z_1^2 z_2 + 2\,\overline{z_2}\,z_1 z_2^2, \frac{2}{3}\,\overline{z_1}^3 z_1 - 2\,\overline{z_1}^2\,\overline{z_2}\,z_2\right\}, \{\overline{z_1}^2 z_1^2 - 4\,\overline{z_1}\,\overline{z_2}\,z_1 z_2 + \overline{z_2}^2 z_2^2, 2\,\overline{z_1}^2 z_1 z_2 - 2\,\overline{z_1}\,\overline{z_2}\,z_2^2\}, \left\{2\,\overline{z_1}^2\,\overline{z_2}\,z_1 - 2\,\overline{z_1}\,\overline{z_2}^2 z_2, 2\,\overline{z_1}\,z_1 z_2^2 - \frac{2}{3}\,\overline{z_2}\,z_2^3\right\}, \left\{\overline{z_1}^2\,\overline{z_2}^2, \frac{2}{3}\,z_1 z_2^3\right\}\right\}$$

```
PsiRegularBasis[3]
```

$$\left\{\{z_1^3, 0\}, \{3\,z_1^2 z_2, 0\}, \{3\,z_1 z_2^2, 0\}, \{z_2^3, 0\}, \left\{\overline{z_2}\,z_1^2, -\frac{\overline{z_1}^3}{3}\right\}, \{-\overline{z_1}\,z_1^2 + 2\,\overline{z_2}\,z_1 z_2, -\overline{z_1}^2\,\overline{z_2}\}, \{2\,\overline{z_1}\,z_1 z_2 - \overline{z_2}\,z_2^2, \overline{z_1}\,\overline{z_2}^2\}, \left\{\overline{z_1}\,z_2^2, \frac{\overline{z_2}^3}{3}\right\}, \{\overline{z_2}^2 z_1, -\overline{z_1}^2 z_2\}, \{-2\,\overline{z_1}\,\overline{z_2}\,z_1 + \overline{z_2}^2 z_2, \overline{z_1}^2 z_1 - 2\,\overline{z_1}\,\overline{z_2}\,z_2\}\right\}$$

```
PsiRegularONBasis[4]
```

$$\{\{\sqrt{5}\,z_1^4, 0\}, \{2\sqrt{5}\,z_1^3 z_2, 0\}, \{\sqrt{30}\,z_1^2 z_2^2, 0\}, \{2\sqrt{5}\,z_1 z_2^3, 0\}, \{\sqrt{5}\,z_2^4, 0\}, \{4\,\overline{z_2}\,z_1^3, -\overline{z_1}^4\}, \{-2\,\overline{z_1}\,z_1^3 + 6\,\overline{z_2}\,z_1^2 z_2, -2\,\overline{z_1}^3\,\overline{z_2}\}, \{2\sqrt{6}\,\overline{z_1}\,z_1^2 z_2 - 2\sqrt{6}\,\overline{z_2}\,z_1 z_2^2, \sqrt{6}\,\overline{z_1}^2\,\overline{z_2}^2\}, \{6\,\overline{z_1}\,z_1 z_2^2 - 2\,\overline{z_2}\,z_2^3, 2\,\overline{z_1}\,\overline{z_2}^3\}, \{4\,\overline{z_1}\,z_2^3, \overline{z_2}^4\}, \{3\sqrt{2}\,\overline{z_2}^2 z_1^2, -2\sqrt{2}\,\overline{z_1}^3 z_2\}, \{-3\sqrt{2}\,\overline{z_1}\,\overline{z_2}\,z_1^2 + 3\sqrt{2}\,\overline{z_2}^2 z_1 z_2, \sqrt{2}\,\overline{z_1}^3 z_1 - 3\sqrt{2}\,\overline{z_1}^2\,\overline{z_2}\,z_2\}, \{\sqrt{3}\,\overline{z_1}^2 z_1^2 - 4\sqrt{3}\,\overline{z_1}\,\overline{z_2}\,z_1 z_2 + \sqrt{3}\,\overline{z_2}^2 z_2^2, 2\sqrt{3}\,\overline{z_1}^2\,\overline{z_2}\,z_1 - 2\sqrt{3}\,\overline{z_1}\,\overline{z_2}^2 z_2\}, \{3\sqrt{2}\,\overline{z_1}^2 z_1 z_2 - 3\sqrt{2}\,\overline{z_1}\,\overline{z_2}\,z_2^2, 3\sqrt{2}\,\overline{z_1}\,\overline{z_2}^2 z_1 - \sqrt{2}\,\overline{z_2}^3 z_2\}, \{3\sqrt{2}\,\overline{z_1}^2 z_2^2, 2\sqrt{2}\,\overline{z_2}^3 z_1\}\}$$

Help

To get the usage message of a package function, evaluate the input **?FunctionName**.

```
n:=Names["RegularHarmonics`*"];ToExpression[Table[StringJoin["?",
ToString[Part[n,i]]],{i,Length[n]}]];
```

BallIntegral[f,z] gives the normalized integral over the unit ball B of the polynomial f[z,z]. The volume of B is assumed to be 1

BallNorm[f,z] gives the normalized L^2 norm over the unit ball B of the polynomial f[z,z]. BallNorm[{f_1,f_2},z] gives the normalized L^2 norm over the unit ball B of the quaternionic polynomial $f_1+f_2 j$. The volume of B is assumed to be 1

BallONBasisP[p,q,z] gives a $L^2(B)$-orthonormal basis of the space $\mathcal{H}_{p,q}$ of the complex harmonic homogeneous polynomials of degree p in z_1, z_2 and q in $\overline{z_1}$, $\overline{z_2}$. It is obtained from a basis introduced by Sudbery (see References). BallONBasisP[k,z] gives a $L^2(B)$-orthonormal basis of the space $\mathcal{H}_k = \bigoplus \mathcal{H}_{p,q}$ of the complex harmonic homogeneous polynomials of degree k

BallProduct[f,g,z] gives the normalized L^2 product over the unit ball B of the complex polynomials f[z,z] and g[z,z]. BallProduct[{f_1,f_2},{g_1,g_2},z] gives the normalized L^2 product over the unit ball B of the quaternionic polynomials $f_1+f_2 j$ and $g_1+g_2 j$. The volume of B is assumed to be 1

BasisP[p,q,z] gives a basis of the space $\mathcal{H}_{p,q}$ of the complex harmonic homogeneous polynomials of degree p in z_1, z_2 and q in $\overline{z_1}$, $\overline{z_2}$. It is a $L^2(S)$-orthogonal basis introduced by Sudbery (see References). BasisP[k,z] gives a basis of the space $\mathcal{H}_k = \bigoplus \mathcal{H}_{p,q}$ of the complex harmonic homogeneous polynomials of degree k

ComplexLaplacian[f, z] gives the complex Laplacian of f with respect to z. In C^2 it is equal to 1/4 of the real Laplacian of f

CRF[{f_1,f_2},z] computes the (left)Cauchy-Riemann-Fueter equations of $f=f_1+f_2 j$ i.e. the pair $\{\frac{\partial f_1}{\partial \overline{z_1}} - \frac{\partial \overline{f_2}}{\partial z_2}, \frac{\partial f_1}{\partial \overline{z_2}} + \frac{\partial \overline{f_2}}{\partial z_1}\}$

CtoH[{f_1,f_2},z,x] converts the pair {f_1,f_2} as a complex function of $z_1=x_0+i x_1$ and $z_2=x_2+i x_3$ to the 4-tuple of the real components of the quaternion $f_1+f_2 j$

CtoR[f,z,x] converts a complex expression f[z,z] as a function of $z_1=x_0+i x_1$ and $z_2=x_2+i x_3$ to the form {real part, imaginary part} in terms of x_0,x_1,x_2,x_3

DbarN[f,z] gives the normal part $\overline{\partial_n} f = z_1 \frac{\partial f}{\partial \overline{z_1}} + z_2 \frac{\partial f}{\partial \overline{z_2}}$ of $\overline{\partial} f$ with respect to the unit sphere S

ExteriorGaussExtension[f,z] gives the harmonic extension on the complement of the unit ball of the restriction of the polynomial f[z,z] to the unit sphere S

GaussExtension[f,z] gives the (polynomial) harmonic extension of the restriction of the polynomial f[z,z] to the unit sphere S

GaussForm[f,z] gives the harmonic representation of the restriction of the polynomial f[z,z] to the unit sphere S.
The output is a list of pairs {h_k,2k}, with h_k harmonic and such that the sum $\sum_k h_k \mid z \mid^{2k}$ is equal to f on S

HtoC[{g_0,g_1,g_2,g_3},x,z] converts the 4-tuple {g_0,g_1,g_2,g_3} as a function of x to the complex pair {$g_0+\mathbb{i}g_1$,$g_2+\mathbb{i}g_3$} as a function of z,z

KelvinTransform[f,z] gives the Kelvin Transform of f[z,z]

L[f,z] applies the Cauchy-Riemann tangential operator $z_2\frac{\partial}{\partial \overline{z_1}}-z_1\frac{\partial}{\partial \overline{z_2}}$ to f

Laplacian[f, x] gives the Laplacian of f with respect to x

Lbar[f,z] applies the conjugate Cauchy-Riemann tangential operator $\overline{z_2}\frac{\partial}{\partial z_1}-\overline{z_1}\frac{\partial}{\partial z_2}$ to f

LeadingTerm[f,z] gives the leading term of the polynomial f[z,z] with respect to the graded lexicographic order with $z_1 > z_2 > \overline{z_1} > \overline{z_2}$

NFueter[f,z] applies the differential operator $N=\overline{z_1}\frac{\partial}{\partial \overline{z_1}}+z_2\frac{\partial}{\partial z_2}$ to f

ONBasisP[p,q,z] gives a L^2 (S)-orthonormal basis of the space $\mathcal{H}_{p,q}$ of the complex harmonic homogeneous polynomials of degree p in z_1, z_2 and q in $\overline{z_1}$, $\overline{z_2}$.
It is obtained from a basis introduced by Sudbery (see References).
ONBasisP[k,z] gives a L^2 (S)-orthonormal basis of the space $\mathcal{H}_k = \bigoplus\mathcal{H}_{p,q}$ of the complex harmonic homogeneous polynomials of degree k

OnS[f,z] computes the restriction of f[z,z] to the unit sphere S in C^2

PsiCRF[{f_1,f_2},z] computes the (left)Cauchy-Riemann-Fueter equations for (left)ψ-regular functions i.e. the pair $\{\frac{\partial f_1}{\partial \overline{z_1}}-\frac{\partial \overline{f_2}}{\partial z_2},\frac{\partial f_1}{\partial \overline{z_2}}+\frac{\partial \overline{f_2}}{\partial z_1}\}$

PsiRegularBallONBasis[k,z] gives a L^2 (B)-orthonormal basis of the right quaternionic module U_k^ψ of the (left) ψ-regular homogeneous polynomials of degree k in z

PsiRegularBasis[k,z] gives a basis of the right quaternionic module U_k^ψ of the (left) ψ-regular homogeneous polynomials of degree k in z.
The restrictions to S gives a basis of the regular harmonics.

PsiRegularExtension[{f_1,f_2},z] gives, if it exists, the (left)ψ-regular extension of the restriction of $f=f_1+f_2j$ to the unit sphere.
PsiRegularExtension[h_1,z] gives a ψ-regular polynomial $f=f_1+f_2j$ such that $f_1=h_1$ on the unit sphere.
Here f_1,f_2 and h_1 must be polynomial functions of z,z

PsiRegularExtensionQ[{f_1,f_2},z] tests for (left)ψ-regularity of the harmonic extension of the restriction of $f=f_1+f_2j$ to the unit sphere. Here f_1 and f_2 must be polynomial functions of z,z

PsiRegularONBasis[k,z] gives a L^2 (S)–orthonormal basis of the right quaternionic module U_k^ψ of the (left) ψ–regular homogeneous polynomials of degree k in z

PsiRegularQ[{f_1,f_2},z] tests for (left)ψ–regularity of f=f_1+f_2j. Here f is a function of z,z

RealBallONBasisP[k,z] gives a L^2 (B)–orthonormal real basis of the space $\mathcal{H}_k$ of the complex harmonic homogeneous polynomials of degree k in z. It is obtained from a complex basis introduced by Sudbery (see References).

RealBasisP[k,z] gives a real basis of the space $\mathcal{H}_k$ of the complex harmonic homogeneous polynomials of degree k in z. It is a L^2 (S)–orthogonal basis obtained from a complex basis introduced by Sudbery (see References).

RealONBasisP[k,z] gives a L^2 (S)–orthonormal real basis of the space $\mathcal{H}_k$ of the complex harmonic homogeneous polynomials of degree k in z. It is obtained from a complex basis introduced by Sudbery (see References).

RegularBallONBasis[k,z] gives a L^2 (B)–orthonormal basis of the right quaternionic module U_k of the (left) regular homogeneous polynomials of degree k in z

RegularBasis[k,z] gives a basis of the right quaternionic module U_k of the (left) regular homogeneous polynomials of degree k in z

RegularExtension[{f_1,f_2},z] gives, if it exists, the (left)regular extension of the restriction of f=f_1+f_2j to the unit sphere. RegularExtension[h_1,z] gives a regular polynomial f=f_1+f_2j such that f_1=h_1 on the unit sphere. Here f_1,f_2 and h_1 must be polynomial functions of z,z

RegularExtensionQ[{f_1,f_2},z] tests for (left)Fueter–regularity of the harmonic extension of the restriction of f=f_1+f_2j to the unit sphere. Here f_1 and f_2 must be polynomial functions of z,z

RegularHarmonics.m is a package that implements computations with (left)regular quaternionic polynomials and harmonic functions of two complex variables.

RegularONBasis[k,z] gives a L^2 (S)–orthonormal basis of the right quaternionic module U_k of the (left) regular homogeneous polynomials of degree k in z

RegularQ[{f_1,f_2},z] tests for (left)Fueter–regularity of f=f_1+f_2j. Here f is a function of z,z

RtoC[{g_1,g_2},x,z] converts the real pair {g_1,g_2} as a function of x_0,x_1,x_2,x_3 to the complex expression $g_1+\mathbb{i}g_2$ as a function of $z_1=x_0+\mathbb{i}x_1$ and $z_2=x_2+\mathbb{i}x_3$

SphereIntegral[f,z] gives the normalized integral over the unit sphere S of the polynomial f[z,z]. The volume of S is assumed to be 1

SphereNorm[f,z] gives the normalized L^2 norm over the unit sphere S of the polynomial f[z,z]. SphereNorm[{f_1,f_2},z] gives the normalized L^2 norm over the unit sphere S of the quaternionic polynomial f_1+f_2j. The volume of S is assumed to be 1

SphereProduct[f,g,z] gives the normalized L^2 product over the unit sphere S of the complex polynomials f[z,z] and g[z,z]. SphereProduct[{f_1,f_2},{g_1,g_2},z] gives the normalized L^2 product over the unit sphere S of the quaternionic polynomials f_1+f_2j and g_1+g_2j. The volume of S is assumed to be 1

TFueter[f,z] applies the tangential operator $T=\overline{z_2}\frac{\partial}{\partial \overline{z_1}}-z_1\frac{\partial}{\partial \overline{z_2}}$ to f

ToComplexNorm[f,z] converts the expression f in terms of the norms of z,z[1],z[2]

If Tonorm has value True, subsequent calls to many functions of the package express results in terms of the norms of z,z[1],z[2]

ToRealNorm[f,x] converts the expression f in terms of the norm of x

TotalDegree[f,z] gives the total degree of a polynomial f in z,z

x is the default indexed real variable with four components x[0],x[1],x[2],x[3]; it represents the quaternion $x_0+ix_1+jx_2+kx_3$

z is the default indexed complex variable in C^2 with two components z[1],z[2]; it represents the quaternion z[1]+z[2]j. Conjugate[z[1]] is input as z–[1] and output as $\overline{z_1}$. The same for $\overline{z_2}$.

References

[S] A. Sudbery, *Quaternionic analysis*, Mat. Proc. Camb. Phil. Soc. vol. 85, 199-225 (1979)

[P1] A. Perotti, *Quaternionic regular functions and the dibar-Neumann problem in C^2*, preprint UTM n.654 (2003) (http://www.science.unitn.it/~perotti/reg.pdf)

[P2] A. Perotti, *A differential criterium for regularity of quaternionic functions*, Comptes Rendus Mathematique, Volume 337, Issue 2, 89-92 (2003)

- http : // www.science.unitn.it / ~ perotti / regular harmonics.htm